

Example 13a The Weyl tensor vanishes in 3d – direct proof

```

1 {x,y,z}::Coordinate.
2 {a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,u,v,w#}::Indices (values={x,y,z},position=independent).
3
4 \partial{#}::PartialDerivative.
5
6 g_{a b}::Metric.
7 g^{a b}::InverseMetric.
8
9 {\partial_{a b}{g_{c d}},\partial_{a}{g_{b c}},g_{a b},g^{a b}}::SortOrder.
10
11 GammaU := \Gamma^{a}_{b c} -> 1/2 g^{a d} ( \partial_{b}{g_{d c}}
12                                     + \partial_{c}{g_{b d}}
13                                     - \partial_{d}{g_{b c}}). # cdb(Gamma.000,GammaU)
14
15 GammaD := \Gamma_{a b c} -> 1/2 ( \partial_{b}{g_{a c}}
16                                 + \partial_{c}{g_{b a}}
17                                 - \partial_{a}{g_{b c}}). # cdb(Gamma.010,GammaD)
18
19 Rabcd := R_{a b c d} -> \partial_{c}{\Gamma_{a b d}}
20                      - \partial_{d}{\Gamma_{a b c}}
21                      + \Gamma_{e a d} \Gamma^{e}_{b c}
22                      - \Gamma_{e a c} \Gamma^{e}_{b d}. # cdb (Rabcd.000,Rabcd)
23
24 Rab      := R_{a b} -> g^{c d} R_{a c b d}. # cdb (Rab.000,Rab)
25
26 Rscalar := R -> g^{a b} R_{a b}. # cdb (R.000,Rscalar)
27
28 # Weyl in 3-dimensions
29
30 Cabcd := R_{a b c d} - (R_{a c} g_{b d} - R_{a d} g_{b c})
31                      - (g_{a c} R_{b d} - g_{a d} R_{b c})
32                      + 1/2 R (g_{a c} g_{b d} - g_{a d} g_{b c}). # cdb (ex-13a.100,Cabcd)
33
34 # Use 8 Cabcd to clear the fractions
35
36 EightCabcd := 8 @(Cabcd). # cdb (ex-13a.110,EightCabcd)

```

```

37
38 substitute      (Cabcd,Rscalar)
39 substitute      (Cabcd,Rab)
40 substitute      (Cabcd,Rabcd)
41 substitute      (Cabcd,GammaU)
42 substitute      (Cabcd,GammaD)
43
44 distribute      (Cabcd)
45
46 sort_product    (Cabcd)
47 rename_dummies  (Cabcd)
48 canonicalise    (Cabcd)                                # cdb (ex-13a.101,Cabcd)
49
50 EightCabcd := 8 @(Cabcd).                                # cdb (ex-13a.111,EightCabcd)
51
52 gab := {g_{x x} = gxx, g_{x y} = gxy, g_{x z} = gxz,
53         g_{y x} = gxy, g_{y y} = gyy, g_{y z} = gyz,
54         g_{z x} = gxz, g_{z y} = gyz, g_{z z} = gzz}.
55
56 complete (gab, $g^{a b}$)
57 evaluate (Cabcd,gab)                                    # cdb (ex-13a.102,Cabcd)
58 evaluate (EightCabcd,gab)                               # cdb (ex-13a.112,EightCabcd)

```

$$8C_{abcd} = 8R_{abcd} - 8R_{ac}g_{bd} + 8R_{ad}g_{bc} - 8g_{ac}R_{bd} + 8g_{ad}R_{bc} + 4R(g_{ac}g_{bd} - g_{ad}g_{bc}) \quad (\text{ex-13a.110})$$

$$\begin{aligned}
&= 4\partial_{bc}g_{ad} - 4\partial_{ac}g_{bd} - 4\partial_{bd}g_{ac} + 4\partial_{ad}g_{bc} + 2\partial_a g_{de}\partial_b g_{cf}g^{ef} + 2\partial_a g_{de}\partial_c g_{bf}g^{ef} - 2\partial_a g_{de}\partial_f g_{bc}g^{ef} + 2\partial_b g_{ce}\partial_d g_{af}g^{ef} + 2\partial_c g_{be}\partial_d g_{af}g^{ef} \\
&\quad - 2\partial_d g_{ae}\partial_f g_{bc}g^{ef} - 2\partial_b g_{ce}\partial_f g_{ad}g^{ef} - 2\partial_c g_{be}\partial_f g_{ad}g^{ef} + 2\partial_e g_{ad}\partial_f g_{bc}g^{ef} - 2\partial_a g_{ce}\partial_b g_{df}g^{ef} - 2\partial_a g_{ce}\partial_d g_{bf}g^{ef} + 2\partial_a g_{ce}\partial_f g_{bd}g^{ef} - 2\partial_b g_{de}\partial_c g_{af}g^{ef} \\
&\quad - 2\partial_c g_{ae}\partial_d g_{bf}g^{ef} + 2\partial_c g_{ae}\partial_f g_{bd}g^{ef} + 2\partial_b g_{de}\partial_f g_{ac}g^{ef} + 2\partial_d g_{be}\partial_f g_{ac}g^{ef} - 2\partial_e g_{ac}\partial_f g_{bd}g^{ef} - 4\partial_{ce}g_{af}g_{bd}g^{ef} + 4\partial_{ac}g_{ef}g_{bd}g^{ef} + 4\partial_{ef}g_{ac}g_{bd}g^{ef} \\
&\quad - 4\partial_{ae}g_{cf}g_{bd}g^{ef} - 2\partial_a g_{ef}\partial_c g_{gh}g_{bd}g^{eg}g^{fh} - 4\partial_e g_{af}\partial_g g_{ch}g_{bd}g^{eg}g^{fh} + 4\partial_e g_{af}\partial_g g_{ch}g_{bd}g^{eh}g^{fg} + 4\partial_a g_{ce}\partial_f g_{gh}g_{bd}g^{eg}g^{fh} - 2\partial_a g_{ce}\partial_f g_{gh}g_{bd}g^{ef}g^{gh} \\
&\quad + 4\partial_c g_{ae}\partial_f g_{gh}g_{bd}g^{eg}g^{fh} - 2\partial_c g_{ae}\partial_f g_{gh}g_{bd}g^{ef}g^{gh} - 4\partial_e g_{ac}\partial_f g_{gh}g_{bd}g^{eg}g^{fh} + 2\partial_e g_{ac}\partial_f g_{gh}g_{bd}g^{ef}g^{gh} + 4\partial_{de}g_{af}g_{bc}g^{ef} - 4\partial_{ad}g_{ef}g_{bc}g^{ef} \\
&\quad - 4\partial_{ef}g_{ad}g_{bc}g^{ef} + 4\partial_{ae}g_{df}g_{bc}g^{ef} + 2\partial_a g_{ef}\partial_d g_{gh}g_{bc}g^{eg}g^{fh} + 4\partial_e g_{af}\partial_g g_{dh}g_{bc}g^{eg}g^{fh} - 4\partial_e g_{af}\partial_g g_{dh}g_{bc}g^{eh}g^{fg} - 4\partial_a g_{de}\partial_f g_{gh}g_{bc}g^{eg}g^{fh} \\
&\quad + 2\partial_a g_{de}\partial_f g_{gh}g_{bc}g^{ef}g^{gh} - 4\partial_d g_{ae}\partial_f g_{gh}g_{bc}g^{eg}g^{fh} + 2\partial_d g_{ae}\partial_f g_{gh}g_{bc}g^{ef}g^{gh} + 4\partial_e g_{ad}\partial_f g_{gh}g_{bc}g^{eg}g^{fh} - 2\partial_e g_{ad}\partial_f g_{gh}g_{bc}g^{ef}g^{gh} - 4\partial_{de}g_{bf}g_{ac}g^{ef} \\
&\quad + 4\partial_{bd}g_{ef}g_{ac}g^{ef} + 4\partial_{ef}g_{bd}g_{ac}g^{ef} - 4\partial_{be}g_{df}g_{ac}g^{ef} - 2\partial_b g_{ef}\partial_d g_{gh}g_{ac}g^{eg}g^{fh} - 4\partial_e g_{bf}\partial_g g_{dh}g_{ac}g^{eg}g^{fh} + 4\partial_e g_{bf}\partial_g g_{dh}g_{ac}g^{eh}g^{fg} + 4\partial_b g_{de}\partial_f g_{gh}g_{ac}g^{eg}g^{fh} \\
&\quad - 2\partial_b g_{de}\partial_f g_{gh}g_{ac}g^{ef}g^{gh} + 4\partial_d g_{be}\partial_f g_{gh}g_{ac}g^{eg}g^{fh} - 2\partial_d g_{be}\partial_f g_{gh}g_{ac}g^{ef}g^{gh} - 4\partial_e g_{bd}\partial_f g_{gh}g_{ac}g^{eg}g^{fh} + 2\partial_e g_{bd}\partial_f g_{gh}g_{ac}g^{ef}g^{gh} + 4\partial_{ce}g_{bf}g_{ad}g^{ef} \\
&\quad - 4\partial_{bc}g_{ef}g_{ad}g^{ef} - 4\partial_{ef}g_{bc}g_{ad}g^{ef} + 4\partial_{be}g_{cf}g_{ad}g^{ef} + 2\partial_b g_{ef}\partial_c g_{gh}g_{ad}g^{eg}g^{fh} + 4\partial_e g_{bf}\partial_g g_{ch}g_{ad}g^{eg}g^{fh} - 4\partial_e g_{bf}\partial_g g_{ch}g_{ad}g^{eh}g^{fg} - 4\partial_b g_{ce}\partial_f g_{gh}g_{ad}g^{eg}g^{fh} \\
&\quad + 2\partial_b g_{ce}\partial_f g_{gh}g_{ad}g^{ef}g^{gh} - 4\partial_c g_{be}\partial_f g_{gh}g_{ad}g^{eg}g^{fh} + 2\partial_c g_{be}\partial_f g_{gh}g_{ad}g^{ef}g^{gh} + 4\partial_e g_{bc}\partial_f g_{gh}g_{ad}g^{eg}g^{fh} - 2\partial_e g_{bc}\partial_f g_{gh}g_{ad}g^{ef}g^{gh} + 4\partial_{ef}g_{gh}g_{ac}g_{bd}g^{eg}g^{fh} \\
&\quad - 4\partial_{ef}g_{gh}g_{ad}g_{bc}g^{eg}g^{fh} - 4\partial_{ef}g_{gh}g_{ac}g_{bd}g^{ef}g^{gh} + 4\partial_{ef}g_{gh}g_{ad}g_{bc}g^{ef}g^{gh} - 2\partial_e g_{fg}\partial_h g_{ij}g_{ac}g_{bd}g^{ei}g^{fh}g^{gj} + 2\partial_e g_{fg}\partial_h g_{ij}g_{ad}g_{bc}g^{ei}g^{fh}g^{gj} \\
&\quad + 3\partial_e g_{fg}\partial_h g_{ij}g_{ac}g_{bd}g^{eh}g^{fi}g^{gj} - 3\partial_e g_{fg}\partial_h g_{ij}g_{ad}g_{bc}g^{eh}g^{fi}g^{gj} - 4\partial_e g_{fg}\partial_h g_{ij}g_{ac}g_{bd}g^{ef}g^{gi}g^{hj} + 4\partial_e g_{fg}\partial_h g_{ij}g_{ad}g_{bc}g^{ef}g^{gi}g^{hj} \\
&\quad + 4\partial_e g_{fg}\partial_h g_{ij}g_{ac}g_{bd}g^{ef}g^{gh}g^{ij} - 4\partial_e g_{fg}\partial_h g_{ij}g_{ad}g_{bc}g^{ef}g^{gh}g^{ij} - \partial_e g_{fg}\partial_h g_{ij}g_{ac}g_{bd}g^{eh}g^{fg}g^{ij} + \partial_e g_{fg}\partial_h g_{ij}g_{ad}g_{bc}g^{eh}g^{fg}g^{ij} \quad (\text{ex-13a.111})
\end{aligned}$$

$$= 0 \quad (\text{ex-13a.112})$$

Example 13b The Weyl tensor vanishes in 3d – orthonormal basis

```

1  {a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).
2
3  g_{a b}::Metric.
4  g^{a b}::InverseMetric.
5
6  R_{a b c d}::RiemannTensor.
7
8  ex{#}::LaTeXForm{"e_x"}.
9  ey{#}::LaTeXForm{"e_y"}.
10 ez{#}::LaTeXForm{"e_z"}.
11
12 {R_{a b c d}, g_{a b}, g^{a b}}::SortOrder.
13
14 Rab      := R_{a b} -> g^{c d} R_{a c b d}.
15
16 Rscalar := R -> g^{a b} R_{a b}.
17
18 gab := g^{a b} -> ex^{a} ex^{b} + ey^{a} ey^{b} + ez^{a} ez^{b}.
19
20 ortho := {ex^{a} ex^{b} g_{a b} -> 1, ey^{a} ey^{b} g_{a b} -> 1, ez^{a} ez^{b} g_{a b} -> 1,
21           ex^{a} ey^{b} g_{a b} -> 0, ex^{a} ez^{b} g_{a b} -> 0,
22           ey^{a} ex^{b} g_{a b} -> 0, ey^{a} ez^{b} g_{a b} -> 0,
23           ez^{a} ex^{b} g_{a b} -> 0, ez^{a} ey^{b} g_{a b} -> 0}.
24
25 # Weyl in 3-dimensions
26
27 Cabcd := R_{a b c d} - (R_{a c} g_{b d} - R_{a d} g_{b c})
28         - (g_{a c} R_{b d} - g_{a d} R_{b c})
29         + 1/2 R (g_{a c} g_{b d} - g_{a d} g_{b c}).    # cdb (ex-13b.100,Cabcd)
30
31
32 substitute (Cabcd, Rscalar)                # cdb(ex-13b.101,Cabcd)
33 substitute (Cabcd, Rab)                    # cdb(ex-13b.102,Cabcd)
34 distribute (Cabcd)                        # cdb(ex-13b.103,Cabcd)
35
36 Cabcd := C_{a b c d} -> @(Cabcd).

```

```

37 Cxyxy := C_{a b c d} ex^{a} ey^{b} ex^{c} ey^{d}.
38                                     # cdb(ex-13b.104,Cxyxy)
39
40 substitute      (Cxyxy,Cabcd)      # cdb(ex-13b.105,Cxyxy)
41 distribute      (Cxyxy)            # cdb(ex-13b.106,Cxyxy)
42
43 substitute      (Cxyxy, ortho, repeat=True) # cdb(ex-13b.107,Cxyxy)
44
45 substitute      (Cxyxy, gab)        # cdb(ex-13b.108,Cxyxy)
46 distribute      (Cxyxy)            # cdb(ex-13b.109,Cxyxy)
47
48 sort_product    (Cxyxy)            # cdb(ex-13b.110,Cxyxy)
49 rename_dummies  (Cxyxy)            # cdb(ex-13b.111,Cxyxy)
50 canonicalise    (Cxyxy)            # cdb(ex-13b.112,Cxyxy)

```

$$\text{ex-13b.101} := R_{abcd} - R_{ac}g_{bd} + R_{ad}g_{bc} - g_{ac}R_{bd} + g_{ad}R_{bc} + \frac{1}{2}g^{ef}R_{ef}(g_{ac}g_{bd} - g_{ad}g_{bc})$$

$$\text{ex-13b.102} := R_{abcd} - g^{ef}R_{aecf}g_{bd} + g^{fe}R_{afde}g_{bc} - g_{ac}g^{fe}R_{bfde} + g_{ad}g^{ef}R_{becf} + \frac{1}{2}g^{ef}g^{gh}R_{egfh}(g_{ac}g_{bd} - g_{ad}g_{bc})$$

$$\text{ex-13b.103} := R_{abcd} - g^{ef}R_{aecf}g_{bd} + g^{fe}R_{afde}g_{bc} - g_{ac}g^{fe}R_{bfde} + g_{ad}g^{ef}R_{becf} + \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ac}g_{bd} - \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ad}g_{bc}$$

$$C_{abcd}e_x^a e_y^b e_x^c e_y^d = \left(R_{abcd} - g^{ef} R_{aecf} g_{bd} + g^{fe} R_{afde} g_{bc} - g_{ac} g^{fe} R_{bfde} + g_{ad} g^{ef} R_{becf} + \frac{1}{2} g^{ef} g^{gh} R_{egfh} g_{ac} g_{bd} - \frac{1}{2} g^{ef} g^{gh} R_{egfh} g_{ad} g_{bc} \right) e_x^a e_y^b e_x^c e_y^d \quad (\text{ex-13b.105})$$

$$= R_{abcd} e_x^a e_y^b e_x^c e_y^d - g^{ef} R_{aecf} g_{bd} e_x^a e_y^b e_x^c e_y^d + g^{fe} R_{afde} g_{bc} e_x^a e_y^b e_x^c e_y^d - g_{ac} g^{fe} R_{bfde} e_x^a e_y^b e_x^c e_y^d + g_{ad} g^{ef} R_{becf} e_x^a e_y^b e_x^c e_y^d + \frac{1}{2} g^{ef} g^{gh} R_{egfh} g_{ac} g_{bd} e_x^a e_y^b e_x^c e_y^d - \frac{1}{2} g^{ef} g^{gh} R_{egfh} g_{ad} g_{bc} e_x^a e_y^b e_x^c e_y^d \quad (\text{ex-13b.106})$$

$$= R_{abcd} e_x^a e_y^b e_x^c e_y^d - g^{ef} R_{aecf} e_x^a e_x^c - g^{fe} R_{bfde} e_y^b e_y^d + \frac{1}{2} g^{ef} g^{gh} R_{egfh} \quad (\text{ex-13b.107})$$

$$= R_{abcd} e_x^a e_y^b e_x^c e_y^d - (e_x^e e_x^f + e_y^e e_y^f + e_z^e e_z^f) R_{aecf} e_x^a e_x^c - (e_x^f e_x^e + e_y^f e_y^e + e_z^f e_z^e) R_{bfde} e_y^b e_y^d + \frac{1}{2} (e_x^e e_x^f + e_y^e e_y^f + e_z^e e_z^f) (e_x^g e_x^h + e_y^g e_y^h + e_z^g e_z^h) R_{egfh} \quad (\text{ex-13b.108})$$

$$= R_{abcd} e_x^a e_y^b e_x^c e_y^d - e_x^e e_x^f R_{aecf} e_x^a e_x^c - e_y^e e_y^f R_{aecf} e_x^a e_x^c - e_z^e e_z^f R_{aecf} e_x^a e_x^c - e_x^f e_x^e R_{bfde} e_y^b e_y^d - e_y^f e_y^e R_{bfde} e_y^b e_y^d - e_z^f e_z^e R_{bfde} e_y^b e_y^d + \frac{1}{2} e_x^e e_x^f e_x^g e_x^h R_{egfh} + \frac{1}{2} e_x^e e_x^f e_y^g e_y^h R_{egfh} + \frac{1}{2} e_x^e e_x^f e_z^g e_z^h R_{egfh} + \frac{1}{2} e_y^e e_y^f e_x^g e_x^h R_{egfh} + \frac{1}{2} e_y^e e_y^f e_y^g e_y^h R_{egfh} + \frac{1}{2} e_y^e e_y^f e_z^g e_z^h R_{egfh} + \frac{1}{2} e_z^e e_z^f e_x^g e_x^h R_{egfh} + \frac{1}{2} e_z^e e_z^f e_y^g e_y^h R_{egfh} + \frac{1}{2} e_z^e e_z^f e_z^g e_z^h R_{egfh} \quad (\text{ex-13b.109})$$

$$= R_{abcd} e_x^a e_x^c e_y^b e_y^d - R_{aecf} e_x^a e_x^c e_y^e e_y^f - R_{aecf} e_x^a e_x^c e_y^e e_y^f - R_{aecf} e_x^a e_x^c e_z^e e_z^f - R_{bfde} e_x^e e_x^f e_y^b e_y^d - R_{bfde} e_y^b e_y^d e_y^e e_y^f - R_{bfde} e_y^b e_y^d e_z^e e_z^f + \frac{1}{2} R_{egfh} e_x^e e_x^f e_x^g e_x^h + \frac{1}{2} R_{egfh} e_x^e e_x^f e_y^g e_y^h + \frac{1}{2} R_{egfh} e_x^e e_x^f e_z^g e_z^h + \frac{1}{2} R_{egfh} e_x^g e_x^h e_y^e e_y^f + \frac{1}{2} R_{egfh} e_y^e e_y^f e_y^g e_y^h + \frac{1}{2} R_{egfh} e_y^e e_y^f e_z^g e_z^h + \frac{1}{2} R_{egfh} e_x^g e_x^h e_z^e e_z^f + \frac{1}{2} R_{egfh} e_y^g e_y^h e_z^e e_z^f + \frac{1}{2} R_{egfh} e_z^g e_z^h e_z^e e_z^f \quad (\text{ex-13b.110})$$

$$= \frac{1}{2} R_{abcd} e_x^a e_x^c e_y^b e_y^d - \frac{1}{2} R_{abcd} e_x^a e_x^c e_x^b e_x^d - \frac{1}{2} R_{abcd} e_x^a e_x^c e_z^b e_z^d - R_{abcd} e_x^d e_x^b e_y^a e_y^c - R_{abcd} e_y^a e_y^c e_y^d e_y^b - R_{abcd} e_y^a e_y^c e_z^d e_z^b + \frac{1}{2} R_{abcd} e_x^b e_x^d e_y^a e_y^c + \frac{1}{2} R_{abcd} e_y^a e_y^c e_z^b e_z^d + \frac{1}{2} R_{abcd} e_x^b e_x^d e_z^a e_z^c + \frac{1}{2} R_{abcd} e_y^b e_y^d e_z^a e_z^c + \frac{1}{2} R_{abcd} e_z^a e_z^c e_z^b e_z^d \quad (\text{ex-13b.111})$$

$$= 0 \quad (\text{ex-13b.112})$$

Example 13c The Weyl tensor vanishes in 3d – orthonormal basis

```
1 Cxyz := C_{a b c d} ex^{a} ey^{b} ex^{c} ez^{d}. # cdb(ex-13c.101,Cxyz)
2
3 substitute (Cxyz,Cabcd) # cdb(ex-13c.102,Cxyz)
4
5 distribute (Cxyz) # cdb(ex-13c.103,Cxyz)
6
7 substitute (Cxyz, ortho, repeat=True) # cdb(ex-13c.104,Cxyz)
8
9 substitute (Cxyz, gab) # cdb(ex-13c.105,Cxyz)
10 distribute (Cxyz) # cdb(ex-13c.106,Cxyz)
11
12 sort_product (Cxyz) # cdb(ex-13c.107,Cxyz)
13 rename_dummies (Cxyz) # cdb(ex-13c.108,Cxyz)
14 canonicalise (Cxyz) # cdb(ex-13c.109,Cxyz)
```


$$C_{abcd}e_x^ae_y^be_x^ce_z^d = \left(R_{abcd} - g^{ef}R_{aecf}g_{bd} + g^{fe}R_{afde}g_{bc} - g_{ac}g^{fe}R_{bfde} + g_{ad}g^{ef}R_{becf} + \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ac}g_{bd} - \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ad}g_{bc} \right) e_x^ae_y^be_x^ce_z^d \quad (\text{ex-13c.102})$$

$$= R_{abcd}e_x^ae_y^be_x^ce_z^d - g^{ef}R_{aecf}g_{bd}e_x^ae_y^be_x^ce_z^d + g^{fe}R_{afde}g_{bc}e_x^ae_y^be_x^ce_z^d - g_{ac}g^{fe}R_{bfde}e_x^ae_y^be_x^ce_z^d + g_{ad}g^{ef}R_{becf}e_x^ae_y^be_x^ce_z^d + \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ac}g_{bd}e_x^ae_y^be_x^ce_z^d - \frac{1}{2}g^{ef}g^{gh}R_{egfh}g_{ad}g_{bc}e_x^ae_y^be_x^ce_z^d \quad (\text{ex-13c.103})$$

$$= R_{abcd}e_x^ae_y^be_x^ce_z^d - g^{fe}R_{bfde}e_y^be_z^d \quad (\text{ex-13c.104})$$

$$= R_{abcd}e_x^ae_y^be_x^ce_z^d - (e_x^fe_x^e + e_y^fe_y^e + e_z^fe_z^e) R_{bfde}e_y^be_z^d \quad (\text{ex-13c.105})$$

$$= R_{abcd}e_x^ae_y^be_x^ce_z^d - e_x^fe_x^eR_{bfde}e_y^be_z^d - e_y^fe_y^eR_{bfde}e_y^be_z^d - e_z^fe_z^eR_{bfde}e_y^be_z^d \quad (\text{ex-13c.106})$$

$$= R_{abcd}e_x^ae_x^ce_y^be_z^d - R_{bfde}e_x^e e_y^fe_y^be_z^d - R_{bfde}e_y^be_y^e e_y^fe_z^d - R_{bfde}e_y^be_z^de_z^e e_z^fe_z^d \quad (\text{ex-13c.107})$$

$$= R_{abcd}e_x^ae_x^ce_y^be_z^d - R_{abcd}e_x^de_x^be_y^ae_z^c - R_{abcd}e_y^ae_y^de_z^be_z^c - R_{abcd}e_y^ae_z^ce_z^de_z^b \quad (\text{ex-13c.108})$$

$$= 0 \quad (\text{ex-13c.109})$$