

Exercise 6.6 More digging around in Cadabra's datastructure

```
1  {\theta, \varphi}::Coordinate.
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).
3
4  \partial{#}::PartialDerivative.
5
6  g^{a b}::InverseMetric. # essential when using complete (gab, $g^{a b}$)
7
8  Gamma := \Gamma^{a}_{f g} -> 1/2 g^{a b} ( \partial_{g}{g_{b f}}
9                                         + \partial_{f}{g_{b g}}
10                                         - \partial_{b}{g_{f g}} ).
11
12  Rabcd := R^{d}_{e f g} -> \partial_{f}{\Gamma^{d}_{e g}}
13                        - \partial_{g}{\Gamma^{d}_{e f}}
14                        + \Gamma^{d}_{b f} \Gamma^{b}_{e g}
15                        - \Gamma^{d}_{b g} \Gamma^{b}_{e f}.
16
17  Rab := R_{a b} -> R^{c}_{c} _{a b}.
18
19  gab := { g_{\theta \theta} = r**2,
20          g_{\varphi \varphi} = r**2 \sin(\theta)**2 }. # cdb(ex-0606.101,gab)
21
22  complete (gab, $g^{a b}$) # cdb(ex-0606.102,gab)
23
24  substitute (Rabcd, Gamma)
25  substitute (Rab, Rabcd)
26
27  evaluate (Gamma, gab, rhsonly=True) # cdb(ex-0606.103,Gamma)
28  evaluate (Rabcd, gab, rhsonly=True) # cdb(ex-0606.104,Rabcd)
29  evaluate (Rab, gab, rhsonly=True) # cdb(ex-0606.105,Rab)
30
31  indcs = Rab[1][2][0][0] # cdb(ex-0606.106,indcs)
32  compt = Rab[1][2][0][1] # cdb(ex-0606.107,compt)
33
34  # cdbBeg(print.0606)
35  print ('Rab = ' + str(Rab.input_form())+'\n') # reveals Cadabra's internal structure for storing Rab
36
```

```
37 print ('Rab[0] = ' + str(Rab[0]))
38 print ('Rab[1] = ' + str(Rab[1])+'\n')
39
40 print ('Rab[1][0] = ' + str(Rab[1][0]))
41 print ('Rab[1][1] = ' + str(Rab[1][1]))
42 print ('Rab[1][2] = ' + str(Rab[1][2])+'\n')
43
44 print ('Rab[1][2][0] = ' + str(Rab[1][2][0]))
45 print ('Rab[1][2][0][0] = ' + str(Rab[1][2][0][0]))
46 print ('Rab[1][2][0][1] = ' + str(Rab[1][2][0][1]))
47 # cdbEnd(print.0606)
```

$$[g_{\theta\theta} = r^2, \quad g_{\varphi\varphi} = r^2(\sin\theta)^2] \quad (\text{ex-0606.101})$$

$$[g_{\theta\theta} = r^2, \quad g_{\varphi\varphi} = r^2(\sin\theta)^2, \quad g^{\theta\theta} = r^{-2}, \quad g^{\varphi\varphi} = (r^2(\sin\theta)^2)^{-1}] \quad (\text{ex-0606.102})$$

$$\Gamma^a_{fg} \rightarrow \square_{fg}^a \begin{cases} \square_{\varphi\theta}^\varphi = (\tan\theta)^{-1} \\ \square_{\theta\varphi}^\varphi = (\tan\theta)^{-1} \\ \square_{\varphi\varphi}^\theta = -\frac{1}{2}\sin(2\theta) \end{cases} \quad (\text{ex-0606.103})$$

$$R^d_{efg} \rightarrow \square_{eg}^d \begin{cases} \square_{\varphi\varphi}^\theta = (\sin\theta)^2 \\ \square_{\theta\varphi}^\varphi = -1 \\ \square_{\varphi\theta}^\theta = -(\sin\theta)^2 \\ \square_{\theta\theta}^\varphi = 1 \end{cases} \quad (\text{ex-0606.104})$$

$$R_{ab} \rightarrow \square_{ab} \begin{cases} \square_{\varphi\varphi} = (\sin\theta)^2 \\ \square_{\theta\theta} = 1 \end{cases} \quad (\text{ex-0606.105})$$

$$R_{\varphi\varphi} = R_{[\varphi, \varphi]} \quad (\text{ex-0606.106})$$

$$= (\sin\theta)^2 \quad (\text{ex-0606.107})$$

```

1 Rab = R_{a b} -> \components_{a b}(\{\varphi, \varphi\} = (\sin(\theta))**2, {\theta, \theta} = 1)
2
3 Rab[0] = R_{a b}
4 Rab[1] = \components_{a b}(\{\varphi, \varphi\} = (\sin(\theta))**2, {\theta, \theta} = 1)
5
6 Rab[1][0] = a
7 Rab[1][1] = b
8 Rab[1][2] = {\varphi, \varphi} = (\sin(\theta))**2, {\theta, \theta} = 1}
9
10 Rab[1][2][0] = {\varphi, \varphi} = (\sin(\theta))**2
11 Rab[1][2][0][0] = {\varphi, \varphi}
12 Rab[1][2][0][1] = (\sin(\theta))**2

```