Exercise 2.1 Using Cadabra's own product rule

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\nabla{#}::Derivative.
\partial{#}::PartialDerivative.

# templates for covariant derivatives

deriv1 := \nabla_{a}{A?^{b}} -> \partial_{a}{A?^{b}}
                              + \Gamma^{b}_{c a} A?^{c}.

deriv2 := \nabla_{a}{A?_{b}} -> \partial_{a}{A?_{b}}
                              - \Gamma^{c}_{b a} A?_{c}.

# create an object

uv := \nabla_{a}{v_{b} u^{b}}
    - \partial_{a}{v_{b} u^{b}}.      # cdb (ex-0201.101,uv)

# apply the rules, then simplify

product_rule   (uv)                   # cdb (ex-0201.102,uv)
substitute     (uv,deriv1)            # cdb (ex-0201.103,uv)
substitute     (uv,deriv2)            # cdb (ex-0201.104,uv)
distribute     (uv)                   # cdb (ex-0201.105,uv)
sort_product   (uv)                   # cdb (ex-0201.106,uv)
rename_dummies (uv)                   # cdb (ex-0201.107,uv)
```

$$\nabla_a\big(v_b u^b\big) \,-\, \partial_a\big(v_b u^b\big) = \nabla_a v_b u^b + v_b \nabla_a u^b \,-\, \partial_a v_b u^b \,-\, v_b \partial_a u^b \tag{ex-0201.102}$$

$$= \nabla_a v_b u^b + v_b \left(\partial_a u^b + \Gamma^b{}_{ca} u^c\right) \,-\, \partial_a v_b u^b \,-\, v_b \partial_a u^b \tag{ex-0201.103}$$

$$= \left(\partial_a v_b \,-\, \Gamma^c{}_{ba} v_c\right) u^b + v_b \left(\partial_a u^b + \Gamma^b{}_{ca} u^c\right) \,-\, \partial_a v_b u^b \,-\, v_b \partial_a u^b \tag{ex-0201.104}$$

$$= -\,\Gamma^c{}_{ba} v_c u^b + v_b \Gamma^b{}_{ca} u^c \tag{ex-0201.105}$$

$$= -\,\Gamma^c{}_{ba} u^b v_c + \Gamma^b{}_{ca} u^c v_b \tag{ex-0201.106}$$

$$= 0 \tag{ex-0201.107}$$

Exercise 2.1 Using hand crafted product rules

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\nabla{#}::Derivative.
\partial{#}::PartialDerivative.

# templates for covariant derivatives

deriv1 := \nabla_{a}{A?^{b}} -> \partial_{a}{A?^{b}}
                              + \Gamma^{b}_{c a} A?^{c}.

deriv2 := \nabla_{a}{A?_{b}} -> \partial_{a}{A?_{b}}
                              - \Gamma^{c}_{b a} A?_{c}.

# tempaltes for product rules

deriv3 := \nabla_{a}{A?_{b} B?^{c}} -> B?^{c} \nabla_{a}{A?_{b}}
                                     + A?_{b} \nabla_{a}{B?^{c}}.

deriv4 := \partial_{a}{A?_{b} B?^{c}} -> B?^{c} \partial_{a}{A?_{b}}
                                       + A?_{b} \partial_{a}{B?^{c}}.

# create an object

uv := \nabla_{a}{v_{b} u^{b}}
    - \partial_{a}{v_{b} u^{b}}.      # cdb (ex-0201.201,uv)

# apply the rules, then simplify

substitute     (uv,deriv3)            # cdb (ex-0201.202,uv)
substitute     (uv,deriv4)            # cdb (ex-0201.203,uv)
substitute     (uv,deriv1)            # cdb (ex-0201.204,uv)
substitute     (uv,deriv2)            # cdb (ex-0201.205,uv)
distribute     (uv)                   # cdb (ex-0201.206,uv)
sort_product   (uv)                   # cdb (ex-0201.207,uv)
rename_dummies (uv)                   # cdb (ex-0201.208,uv)
```

$$\nabla_a\big(v_b u^b\big) \,-\, \partial_a\big(v_b u^b\big) = u^b \nabla_a v_b + v_b \nabla_a u^b \,-\, \partial_a\big(v_b u^b\big) \tag{ex-0201.202}$$

$$= u^b \nabla_a v_b + v_b \nabla_a u^b \,-\, u^b \partial_a v_b \,-\, v_b \partial_a u^b \tag{ex-0201.203}$$

$$= u^b \nabla_a v_b + v_b \left(\partial_a u^b + \Gamma^b{}_{ca} u^c\right) \,-\, u^b \partial_a v_b \,-\, v_b \partial_a u^b \tag{ex-0201.204}$$

$$= u^b \left(\partial_a v_b \,-\, \Gamma^c{}_{ba} v_c\right) + v_b \left(\partial_a u^b + \Gamma^b{}_{ca} u^c\right) \,-\, u^b \partial_a v_b \,-\, v_b \partial_a u^b \tag{ex-0201.205}$$

$$= -\, u^b \Gamma^c{}_{ba} v_c + v_b \Gamma^b{}_{ca} u^c \tag{ex-0201.206}$$

$$= -\, \Gamma^c{}_{ba} u^b v_c + \Gamma^b{}_{ca} u^c v_b \tag{ex-0201.207}$$

$$= 0 \tag{ex-0201.208}$$