

## Exercise 3.6 Commutation of $\nabla$ on the Riemann tensor – simple computation

```

1  {a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).
2
3  DD{#}::Derivative.
4  \nabla{#}::Derivative.
5
6  RabcdF := R_{a b c d} -> A_{a} B_{b} C_{c} D_{d}.      # cdb(RabcdF.000,RabcdF)
7  RabcdB := A_{a} B_{b} C_{c} D_{d} -> R_{a b c d}.      # cdb(RabcdB.000,RabcdB)
8
9  derivDD := DD_{b c}{V?_{a}} -> R^{d}_{a b c} V?_{d}.    # cdb(derivDD.000,derivDD)
10
11  nablaDD := \nabla_{f}{\nabla_{e}{R_{a b c d}}}
12            - \nabla_{e}{\nabla_{f}{R_{a b c d}}} -> DD_{e f}{R_{a b c d}}.
13
14  # product rule for DD acting on A_{a} B_{b} C_{c} D_{d}
15  pruleDD := DD_{e f}{A_{a} B_{b} C_{c} D_{d}} -> DD_{e f}{A_{a}} B_{b} C_{c} D_{d}
16            + A_{a} DD_{e f}{B_{b}} C_{c} D_{d}
17            + A_{a} B_{b} DD_{e f}{C_{c}} D_{d}
18            + A_{a} B_{b} C_{c} DD_{e f}{D_{d}}.
19            # cdb(pruleDD.000,pruleDD)
20
21  expr := \nabla_{f}{\nabla_{e}{R_{a b c d}}}
22          - \nabla_{e}{\nabla_{f}{R_{a b c d}}}.      # cdb (ex-0306.100, expr)
23
24  substitute (expr,nablaDD)                  # cdb (ex-0306.101, expr)
25  substitute (expr,RabcdF)                  # cdb (ex-0306.102, expr)
26  substitute (expr,pruleDD)                 # cdb (ex-0306.103, expr)
27  substitute (expr,derivDD)                 # cdb (ex-0306.104, expr)
28  sort_product (expr)                       # cdb (ex-0306.105, expr)
29  substitute (expr,RabcdB)                  # cdb (ex-0306.106, expr)

```

$$\nabla_f (\nabla_e R_{abcd}) - \nabla_e (\nabla_f R_{abcd}) = DD_{ef} R_{abcd} \quad (\text{ex-0306.101})$$

$$= DD_{ef} (A_a B_b C_c D_d) \quad (\text{ex-0306.102})$$

$$= DD_{ef} A_a B_b C_c D_d + A_a DD_{ef} B_b C_c D_d + A_a B_b DD_{ef} C_c D_d + A_a B_b C_c DD_{ef} D_d \quad (\text{ex-0306.103})$$

$$= R^g_{aef} A_g B_b C_c D_d + A_a R^g_{bef} B_g C_c D_d + A_a B_b R^g_{cef} C_g D_d + A_a B_b C_c R^g_{def} D_g \quad (\text{ex-0306.104})$$

$$= A_g B_b C_c D_d R^g_{aef} + A_a B_g C_c D_d R^g_{bef} + A_a B_b C_g D_d R^g_{cef} + A_a B_b C_c D_g R^g_{def} \quad (\text{ex-0306.105})$$

$$= R_{ghcd} R^g_{aef} + R_{agcd} R^g_{bef} + R_{abgd} R^g_{cef} + R_{abcg} R^g_{def} \quad (\text{ex-0306.106})$$