

Exercise 4.3 Polynomial products

```
1 {a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).
2
3
4 def get_term (poly,n):
5
6     x^{a}::Weight(label=xnum).      # assign weights to x^{a}
7
8     foo := @(poly).                  # make a copy of poly
9     bah = Ex("xnum = " + str(n))    # choose a target
10    keep_weight (foo,bah)             # extract the target
11
12    return foo
13
14 def poly_product (p,q,n):
15
16    pq = Ex("0")
17
18    for i in range (0,n+1):
19        for j in range (0,i+1):
20            termA = get_term (p,j)
21            termB = get_term (q,i-j)
22            termAB := @(termA) @(termB).
23            pq = pq + termAB
24
25    sort_product (pq)
26    rename_dummies (pq)
27    factor_out (pq,$x^{a?}$)
28
29    return pq
30
31 # -----
32
33 # two polynomials
34
35 polyA := c^{a}
36         + c^{a}_{b} x^b
```

```

37      + c^{a}_{b c} x^b x^c
38      + c^{a}_{b c d} x^b x^c x^d
39      + c^{a}_{b c d e} x^b x^c x^d x^e.      # cdb(ex-0403.100,polyA)
40
41 polyB := d^{f}
42      + d^{f}_{b} x^b
43      + d^{f}_{b c} x^b x^c
44      + d^{f}_{b c d} x^b x^c x^d
45      + d^{f}_{b c d e} x^b x^c x^d x^e.      # cdb(ex-0403.101,polyB)
46
47 # multiply polynomials and truncate
48
49 polyAB = poly_product (polyA,polyB,3)      # cdb(ex-0403.102,polyAB)

```

$$p = c^a + c^a_b x^b + c^a_{bc} x^b x^c + c^a_{bcd} x^b x^c x^d + c^a_{bcde} x^b x^c x^d x^e \quad (\text{ex-0403.100})$$

$$q = d^f + d^f_b x^b + d^f_{bc} x^b x^c + d^f_{bcd} x^b x^c x^d + d^f_{bcde} x^b x^c x^d x^e \quad (\text{ex-0403.101})$$

$$pq = c^a d^f + x^b (c^a d^f_b + c^a_b d^f) + x^b x^c (c^a d^f_{bc} + c^a_b d^f_c + c^a_{bc} d^f) + x^b x^c x^d (c^a d^f_{bcd} + c^a_b d^f_{cd} + c^a_{bc} d^f_d + c^a_{bcd} d^f) \quad (\text{ex-0403.102})$$