

Example 6-01 Evaluating components

```
1  {\theta, \varphi}::Coordinate.  
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).  
3  
4  \partial{#}::PartialDerivative.  
5  
6  V := { V_{\theta} = \varphi, V_{\varphi} = \sin(\theta) }.      # cdb(ex-06.100,V)  
7  dV := \partial_{b}{V_{a}} - \partial_{a}{V_{b}}.              # cdb(ex-06.101,dV)  
8  
9  evaluate (dV, V)    # cdb(ex-06.102,dV)
```

$$V_a = [V_\theta = \varphi, V_\varphi = \sin \theta] \quad (\text{ex-06.100})$$

$$\partial_b V_a - \partial_a V_b = \square_{ab} \begin{cases} \square_{\varphi\theta} = \cos \theta - 1 \\ \square_{\theta\varphi} = -\cos \theta + 1 \end{cases} \quad (\text{ex-06.102})$$

Example 6-02 Ricci tensor of a 2-sphere

```

1  {\theta, \varphi}::Coordinate.
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).
3
4  \partial{#}::PartialDerivative.
5
6  Gamma := \Gamma^{a}_{f g} -> 1/2 g^{a b} ( \partial_{g}{g_{b f}}
7                                     + \partial_{f}{g_{b g}}
8                                     - \partial_{b}{g_{f g}} ).
9
10 Rabcd := R^{d}_{e f g} -> \partial_{f}{\Gamma^{d}_{e g}}
11                        - \partial_{g}{\Gamma^{d}_{e f}}
12                        + \Gamma^{d}_{b f} \Gamma^{b}_{e g}
13                        - \Gamma^{d}_{b g} \Gamma^{b}_{e f}.
14
15 Rab := R_{a b} -> R^{c}_{c} _{a b}.
16
17 gab := { g_{\theta\theta} = r**2,
18          g_{\varphi\varphi} = r**2 \sin(\theta)**2 }. # cdb(ex-06.201,gab)
19
20 iab := { g^{\theta\theta} = 1/r**2,
21          g^{\varphi\varphi} = 1/(r**2 \sin(\theta)**2) }. # cdb(ex-06.202,iab)
22
23 substitute (Rabcd, Gamma)
24 substitute (Rab, Rabcd)
25
26 evaluate (Gamma, gab+iab, rhsonly=True) # cdb(ex-06.203,Gamma)
27 evaluate (Rabcd, gab+iab, rhsonly=True) # cdb(ex-06.204,Rabcd)
28 evaluate (Rab, gab+iab, rhsonly=True) # cdb(ex-06.205,Rab)
29
30 Riem := R^{d}_{e f g}.
31 substitute (Riem, Rabcd)
32 evaluate (Riem, gab+iab) # cdb(ex-06.206,Riem)
33
34 expr := R_{a b}.
35 substitute (expr, Rab)
36 evaluate (expr, gab+iab)

```

```

37
38 # cdbBeg(print.ex-06.02)
39 print ('Rab = ' + str(Rab.input_form())+'\n') # reveals Cadabra's internal structure for storing Rab
40
41 print ('Rab[0] = ' + str(Rab[0]))
42 print ('Rab[1] = ' + str(Rab[1])+'\n')
43
44 print ('Rab[1][0] = ' + str(Rab[1][0]))
45 print ('Rab[1][1] = ' + str(Rab[1][1]))
46 print ('Rab[1][2] = ' + str(Rab[1][2])+'\n')
47
48 print ('Rab[1][2][0] = ' + str(Rab[1][2][0]))
49 print ('Rab[1][2][0][0] = ' + str(Rab[1][2][0][0]))
50 print ('Rab[1][2][0][1] = ' + str(Rab[1][2][0][1]))
51 # cdbEnd(print.ex-06.02)

```

$$[g_{\theta\theta} = r^2, g_{\varphi\varphi} = r^2(\sin\theta)^2] \quad (\text{ex-06.201})$$

$$[g^{\theta\theta} = r^{-2}, g^{\varphi\varphi} = (r^2(\sin\theta)^2)^{-1}] \quad (\text{ex-06.202})$$

$$\Gamma^a_{fg} \rightarrow \square_{fg}^a \begin{cases} \square_{\varphi\theta}^\varphi = (\tan\theta)^{-1} \\ \square_{\theta\varphi}^\varphi = (\tan\theta)^{-1} \\ \square_{\varphi\varphi}^\theta = -\frac{1}{2}\sin(2\theta) \end{cases} \quad (\text{ex-06.203})$$

$$R^d_{efg} \rightarrow \square_{eg}^d{}_f \begin{cases} \square_{\varphi\varphi}^\theta = (\sin\theta)^2 \\ \square_{\theta\varphi}^\varphi = -1 \\ \square_{\varphi\theta}^\theta = -(\sin\theta)^2 \\ \square_{\theta\theta}^\varphi = 1 \end{cases} \quad (\text{ex-06.204})$$

$$R_{ab} \rightarrow \square_{ab} \begin{cases} \square_{\varphi\varphi} = (\sin\theta)^2 \\ \square_{\theta\theta} = 1 \end{cases} \quad (\text{ex-06.205})$$

$$\square_{eg}^d{}_f \begin{cases} \square_{\varphi\varphi}^\theta = (\sin\theta)^2 \\ \square_{\theta\varphi}^\varphi = -1 \\ \square_{\varphi\theta}^\theta = -(\sin\theta)^2 \\ \square_{\theta\theta}^\varphi = 1 \end{cases} \quad (\text{ex-06.206})$$

```

1 Rab = R_{a b} -> \components_{a b}({\varphi, \varphi} = (\sin(\theta))*2, {\theta, \theta} = 1)
2
3 Rab[0] = R_{a b}
4 Rab[1] = \components_{a b}({\varphi, \varphi} = (\sin(\theta))*2, {\theta, \theta} = 1)
5
6 Rab[1][0] = a
7 Rab[1][1] = b
8 Rab[1][2] = {\varphi, \varphi} = (\sin(\theta))*2, {\theta, \theta} = 1}
9
10 Rab[1][2][0] = {\varphi, \varphi} = (\sin(\theta))*2
11 Rab[1][2][0][0] = {\varphi, \varphi}
12 Rab[1][2][0][1] = (\sin(\theta))*2

```

Example 6-03 Using *complete* to compute the inverse metric

This version uses `complete` to compute the inverse metric.

```
1  {\theta, \varphi}::Coordinate.
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).
3
4  \partial{#}::PartialDerivative.
5
6  g^{a b}::InverseMetric. # essential when using complete (gab, $g^{a b}$)
7
8  Gamma := \Gamma^{a}_{f g} -> 1/2 g^{a b} ( \partial_{g}{g_{b f}}
9                                         + \partial_{f}{g_{b g}}
10                                         - \partial_{b}{g_{f g}} ).
11
12  Rabcd := R^{d}_{e f g} -> \partial_{f}{\Gamma^{d}_{e g}}
13                        - \partial_{g}{\Gamma^{d}_{e f}}
14                        + \Gamma^{d}_{b f} \Gamma^{b}_{e g}
15                        - \Gamma^{d}_{b g} \Gamma^{b}_{e f}.
16
17  Rab := R_{a b} -> R^{c}_{c} _{a b}.
18
19  gab := { g_{\theta\theta} = r**2,
20          g_{\varphi\varphi} = r**2 \sin(\theta)**2 }. # cdb(ex-06.301,gab)
21
22  complete (gab, $g^{a b}$) # cdb(ex-06.302,gab)
23
24  substitute (Rabcd, Gamma)
25  substitute (Rab, Rabcd)
26
27  evaluate (Gamma, gab, rhsonly=True) # cdb(ex-06.303,Gamma)
28  evaluate (Rabcd, gab, rhsonly=True) # cdb(ex-06.304,Rabcd)
29  evaluate (Rab, gab, rhsonly=True) # cdb(ex-06.305,Rab)
30
31  Riem := R^{d}_{e f g}.
32  substitute (Riem, Rabcd)
33  evaluate (Riem, gab) # cdb(ex-06.306,Riem)
```

$$[g_{\theta\theta} = r^2, \ g_{\varphi\varphi} = r^2(\sin \theta)^2] \quad (\text{ex-06.301})$$

$$\left[g_{\theta\theta} = r^2, \ g_{\varphi\varphi} = r^2(\sin \theta)^2, \ g^{\theta\theta} = r^{-2}, \ g^{\varphi\varphi} = (r^2(\sin \theta)^2)^{-1} \right] \quad (\text{ex-06.302})$$

$$\Gamma^a_{fg} \rightarrow \square_{fg}^a \begin{cases} \square_{\varphi\theta}^{\varphi} = (\tan \theta)^{-1} \\ \square_{\theta\varphi}^{\varphi} = (\tan \theta)^{-1} \\ \square_{\varphi\varphi}^{\theta} = -\frac{1}{2} \sin(2\theta) \end{cases} \quad (\text{ex-06.303})$$

$$R^d_{efg} \rightarrow \square_{eg}^d \begin{cases} \square_{\varphi\varphi}^{\theta} = (\sin \theta)^2 \\ \square_{\theta\varphi}^{\varphi} = -1 \\ \square_{\varphi\theta}^{\theta} = -(\sin \theta)^2 \\ \square_{\theta\theta}^{\varphi} = 1 \end{cases} \quad (\text{ex-06.304})$$

$$R_{ab} \rightarrow \square_{ab} \begin{cases} \square_{\varphi\varphi} = (\sin \theta)^2 \\ \square_{\theta\theta} = 1 \end{cases} \quad (\text{ex-06.305})$$

$$\square_{eg}^d \begin{cases} \square_{\varphi\varphi}^{\theta} = (\sin \theta)^2 \\ \square_{\theta\varphi}^{\varphi} = -1 \\ \square_{\varphi\theta}^{\theta} = -(\sin \theta)^2 \\ \square_{\theta\theta}^{\varphi} = 1 \end{cases} \quad (\text{ex-06.306})$$

Example 6-04 Components by scalar projection

This example shows how one component of the metric tensor can be computed using a scalar projection.

```

1  {\theta, \varphi}::Coordinate.
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).
3
4  theta{#}::LaTeXForm{"\theta"}.
5  varphi{#}::LaTeXForm{"\varphi"}.
6
7  gab := { g_{\theta \theta} = r**2,
8           g_{\varphi \varphi} = r**2 \sin(\theta)**2 }. # cdb(ex-06.400,gab)
9
10 # obtain components by contracting with basis vectors
11
12 basis := {\theta^{\theta} = 1, \varphi^{\varphi} = 1}.
13
14 compt := g_{a b} \varphi^a \varphi^b. # cdb(ex-06.401,compt)
15
16 evaluate (compt,gab+basis) # cdb(ex-06.402,compt)
17
18 ghiphi = compt._sympy_()
19
20 # cdbBeg(print.ex-06.04)
21 print ('type compt = ' + str(type(compt))) # shows that compt is a Cadabra object
22 print ('type ghiphi = ' + str(type(ghiphi))) # shows that ghiphi is a Python object
23 print ('      compt = ' + str(compt)) # will contain LaTeX markup
24 print ('      ghiphi = ' + str(ghiphi)) # will be pure Python/SymPy
25 # cdbEnd(print.ex-06.04)
26
27 checkpoint.append (compt)

```

$$g_{\varphi\varphi} = g_{ab}\varphi^a\varphi^b \quad (\text{ex-06.401})$$

$$= r^2(\sin\theta)^2 \quad (\text{ex-06.402})$$

```
1  type compt      = <class 'cadabra2.Ex'>
2  type gphiphi    = <class 'sympy.core.mul.Mul'>
3      compt      = (r)**2 (\sin(\theta))**2
4      gphiphi    = r**2*sin(theta)**2
```


Example 6-05 Components by selection

This example shows how one component of the metric tensor can be computed by indexing the result of a call to `evaluate`.

```
1  {\theta, \varphi}::Coordinate.
2  {a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).
3
4  theta{#}::LaTeXForm{"\theta"}.
5  varphi{#}::LaTeXForm{"\varphi"}.
6
7  gab := { g_{\theta \theta}    = r**2,
8           g_{\varphi \varphi} = r**2 \sin(\theta)**2 }.  # cdb(ex-06.500,gab)
9
10 metric := g_{a b}.
11
12 evaluate (metric,gab)
13
14 indcs = metric[2][1][0]          # cdb(ex-06.501,indcs)
15 compt = metric[2][1][1]         # cdb(ex-06.502,compt)
16
17 # cdbBeg(print.ex-06.05)
18 print ('metric = ' + str(metric.input_form())+'\n')  # reveals Cadabra's internal structure for storing metric
19
20 print ('metric[0] = ' + str(metric[0]))
21 print ('metric[1] = ' + str(metric[1]))
22 print ('metric[2] = ' + str(metric[2])+'\n')
23
24 print ('metric[2][1] = '+ str(metric[2][1]))
25 print ('metric[2][1][0] = '+ str(metric[2][1][0]))
26 print ('metric[2][1][1] = '+ str(metric[2][1][1]))
27 # cdbEnd(print.ex-06.05)
28
29 checkpoint.append (indcs)
30 checkpoint.append (compt)
```

$$g_{\varphi\varphi} = g_{[\varphi, \varphi]} \quad (\text{ex-06.501})$$

$$= r^2 (\sin \theta)^2 \quad (\text{ex-06.502})$$

```

1  metric = \components_{a b}({{\theta}, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2)
2
3  metric[0] = a
4  metric[1] = b
5  metric[2] = {{{\theta}, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2}
6
7  metric[2][1] = {\varphi, \varphi} = (r)**2 (\sin(\theta))**2
8  metric[2][1][0] = {\varphi, \varphi}
9  metric[2][1][1] = (r)**2 (\sin(\theta))**2

```