# Exercise 4.3 Polynomial products

```
1
2  {a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).
3
4  def get_term (poly,n):
5
6      x^{a}::Weight(label=xnum).      # assign weights to x^{a}
7
8      foo := @(poly).                 # make a copy of poly
9      bah  = Ex("xnum = " + str(n))   # choose a target
10     keep_weight (foo,bah)           # extract the target
11
12     return foo
13
14 def poly_product (p,q,n):
15
16     pq = Ex("0")
17
18     for i in range (0,n+1):
19        for j in range (0,i+1):
20           termA = get_term (p,j)
21           termB = get_term (q,i-j)
22           termAB := @(termA) @(termB).
23           pq = pq + termAB
24
25     sort_product   (pq)
26     rename_dummies (pq)
27     factor_out     (pq,$x^{a?}$)
28
29     return pq
30
31 # ----------------------------------------------------------------
32
33 # two polynomials
34
35 polyA := c^{a}
36        + c^{a}_{b} x^b
```

```
        + c^{a}_{b c} x^b x^c
        + c^{a}_{b c d} x^b x^c x^d
        + c^{a}_{b c d e} x^b x^c x^d x^e.    # cdb(ex-0403.100,polyA)

polyB := d^{f}
        + d^{f}_{b} x^b
        + d^{f}_{b c} x^b x^c
        + d^{f}_{b c d} x^b x^c x^d
        + d^{f}_{b c d e} x^b x^c x^d x^e.    # cdb(ex-0403.101,polyB)

# multiply polynomials and truncate

polyAB = poly_product (polyA,polyB,3)        # cdb(ex-0403.102,polyAB)
```

$$p = c^{a} + c^{a}{}_{b} x^{b} + c^{a}{}_{bc} x^{b} x^{c} + c^{a}{}_{bcd} x^{b} x^{c} x^{d} + c^{a}{}_{bcde} x^{b} x^{c} x^{d} x^{e} \tag{ex-0403.100}$$

$$q = d^{f} + d^{f}{}_{b} x^{b} + d^{f}{}_{bc} x^{b} x^{c} + d^{f}{}_{bcd} x^{b} x^{c} x^{d} + d^{f}{}_{bcde} x^{b} x^{c} x^{d} x^{e} \tag{ex-0403.101}$$

$$pq = c^{a} d^{f} + x^{b} \left( c^{a} d^{f}{}_{b} + c^{a}{}_{b} d^{f} \right) + x^{b} x^{c} \left( c^{a} d^{f}{}_{bc} + c^{a}{}_{b} d^{f}{}_{c} + c^{a}{}_{bc} d^{f} \right) + x^{b} x^{c} x^{d} \left( c^{a} d^{f}{}_{bcd} + c^{a}{}_{b} d^{f}{}_{cd} + c^{a}{}_{bc} d^{f}{}_{d} + c^{a}{}_{bcd} d^{f} \right) \tag{ex-0403.102}$$