# Exercise 6.5 Digging into Cadabra's datastructure

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

theta{#}::LaTeXForm{"\theta"}.
varphi{#}::LaTeXForm{"\varphi"}.

gab := { g_{\theta \theta}   = r**2,
         g_{\varphi \varphi} = r**2 \sin(\theta)**2 }.   # cdb(ex-0605.100,gab)

metric := g_{a b} -> g_{a b}.  # a trivial rule :)

evaluate (metric,gab,rhsonly=True)

indcs = metric[1][2][1][0]                               # cdb(ex-0605.101,indcs)
compt = metric[1][2][1][1]                               # cdb(ex-0605.102,compt)

# cdbBeg(print.0605)
print ('metric = ' + str(metric.input_form())+'\n')  # reveals Cadabra's internal structure for storing metric

print ('metric[0] = ' + str(metric[0]))
print ('metric[1] = ' + str(metric[1])+'\n')

print ('metric[1][0] = ' + str(metric[1][0]))
print ('metric[1][1] = ' + str(metric[1][1]))
print ('metric[1][2] = ' + str(metric[1][2])+'\n')

print ('metric[1][2][1] = '+ str(metric[1][2][1]))
print ('metric[1][2][1][0] = '+ str(metric[1][2][1][0]))
print ('metric[1][2][1][1] = '+ str(metric[1][2][1][1]))
# cdbEnd(print.0605)
```

$$[\varphi,\ \varphi] \tag{ex-0605.101}$$

$$r^2(\sin\theta)^2 \tag{ex-0605.102}$$

$$\begin{aligned} g_{\varphi\varphi} &= g_{[\varphi,\ \varphi]} && \text{(ex-0605.101)} \\ &= r^2(\sin\theta)^2 && \text{(ex-0605.102)} \end{aligned}$$

```
metric = g_{a b} -> \components_{a b}({{\theta, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2})

metric[0] = g_{a b}
metric[1] = \components_{a b}({{\theta, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2})

metric[1][0] = a
metric[1][1] = b
metric[1][2] = {{\theta, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2}

metric[1][2][1] = {\varphi, \varphi} = (r)**2 (\sin(\theta))**2
metric[1][2][1][0] = {\varphi, \varphi}
metric[1][2][1][1] = (r)**2 (\sin(\theta))**2
```