Exercise 2.2 Covariant derivative of v_{ab}

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\nabla{#}::Derivative.
\partial{#}::PartialDerivative.

# template for covariant derivative of a vector

derivU := \nabla_{a}{A?^{b}} -> \partial_{a}{A?^{b}} + \Gamma^{b}_{c a} A?^{c}.
derivD := \nabla_{a}{A?_{b}} -> \partial_{a}{A?_{b}} - \Gamma^{c}_{b a} A?_{c}.

vab := v_{a b} -> A_{a} B_{b}.
iab := A_{a} B_{b} -> v_{a b}.

pab := \partial_{a}{A_{b}} B_{c} -> \partial_{a}{A_{b} B_{c}} - A_{b} \partial_{a}{B_{c}}.

# create an object

Dvab := \nabla_{a}{v_{b c}}.     # cdb (ex-0202.101,Dvab)

# apply the rule, then simplify

substitute     (Dvab,vab)        # cdb (ex-0202.102,Dvab)
product_rule   (Dvab)            # cdb (ex-0202.103,Dvab)
substitute     (Dvab,derivD)     # cdb (ex-0202.104,Dvab)
substitute     (Dvab,derivU)     # cdb (ex-0202.105,Dvab)
distribute     (Dvab)            # cdb (ex-0202.106,Dvab)
substitute     (Dvab,pab)        # cdb (ex-0202.107,Dvab)
canonicalise   (Dvab)            # cdb (ex-0202.108,Dvab)
substitute     (Dvab,iab)        # cdb (ex-0202.109,Dvab)
sort_product   (Dvab)            # cdb (ex-0202.110,Dvab)
```

$$\nabla_a v_{bc} = \nabla_a (A_b B_c) \tag{ex-0202.102}$$

$$= \nabla_a A_b B_c + A_b \nabla_a B_c \tag{ex-0202.103}$$

$$= \left(\partial_a A_b - \Gamma^d{}_{ba} A_d\right) B_c + A_b \left(\partial_a B_c - \Gamma^d{}_{ca} B_d\right) \tag{ex-0202.104}$$

$$= \left(\partial_a A_b - \Gamma^d{}_{ba} A_d\right) B_c + A_b \left(\partial_a B_c - \Gamma^d{}_{ca} B_d\right) \tag{ex-0202.105}$$

$$= \partial_a A_b B_c - \Gamma^d{}_{ba} A_d B_c + A_b \partial_a B_c - A_b \Gamma^d{}_{ca} B_d \tag{ex-0202.106}$$

$$= \partial_a (A_b B_c) - \Gamma^d{}_{ba} A_d B_c - A_b \Gamma^d{}_{ca} B_d \tag{ex-0202.107}$$

$$= \partial_a (A_b B_c) - \Gamma^d{}_{ba} A_d B_c - A_b \Gamma^d{}_{ca} B_d \tag{ex-0202.108}$$

$$= \partial_a v_{bc} - \Gamma^d{}_{ba} v_{dc} - v_{bd} \Gamma^d{}_{ca} \tag{ex-0202.109}$$

$$= \partial_a v_{bc} - \Gamma^d{}_{ba} v_{dc} - \Gamma^d{}_{ca} v_{bd} \tag{ex-0202.110}$$