

## Example 15 Verifying the BSSN equations

This is short example verifies two of the main equations in the Phys Rev D paper by Miguel Alcubierre, Bernd Bruggmann et al. (Phys.Rev.D. (62) 044034 (2000)).

The code for the full set of BSSN equations can be found at <https://github.com/leo-brewin/adm-bssn-equations>

```
1 {a,b,c,d,e,f,i,j,k,l,m,n,o,p,q,r,s,u#}::Indices(position=independent,values={t,x,y,z}).
2 {t,x,y,z}::Coordinate.
3
4 \partial{#}::PartialDerivative.
5 D{#}::Derivative.
6 DBar{#}::Derivative.
7
8 N::Depends(t,x,y,z).
9
10 g_{a b}::Symmetric.
11 g^{a b}::Symmetric.
12 g_{a}^{b}::KroneckerDelta.
13 g^{a}_{b}::KroneckerDelta.
14
15 g_{a b}::Depends(t,x,y,z).
16 g^{a b}::Depends(t,x,y,z).
17
18 gBar_{a b}::Symmetric.
19 gBar^{a b}::Symmetric.
20 gBar_{a}^{b}::KroneckerDelta.
21 gBar^{a}_{b}::KroneckerDelta.
22
23 gBar_{a b}::Depends(t,x,y,z).
24 gBar^{a b}::Depends(t,x,y,z).
25
26 trK::LaTeXForm("K").
27 detg::LaTeXForm("g").
28 ABar{#}::LaTeXForm("\bar{A}").
29 DBar{#}::LaTeXForm("\bar{D}").
```

## 15.1 Evolution equation for $\phi$

```
1  phi      := \phi -> (1/12) \log(detg).
2  gdotK    := g^{i j} K_{i j} -> trK.
3  DdetgDt  := \partial_t\{detg\} -> detg g^{i j} \partial_t\{g_{i j}\}.
4
5  DgijDt   := \partial_t\{g_{i j}\} -> -2 N K_{i j}.
6
7  dlog      := \partial_{a?}\{\log(A?)\} -> (1/A?)\partial_{a?}\{A?\}.
8  dexp      := \partial_{a?}\{\exp(A?)\} -> \exp(A?)\partial_{a?}\{A?\}.
9
10 dotphi    := \partial_t\{\phi\}.
11
12 substitute (dotphi, phi)           # cdb (ex-15-02.101,dotphi)
13 substitute (dotphi, dlog)          # cdb (ex-15-02.102,dotphi)
14 substitute (dotphi, DdetgDt)       # cdb (ex-15-02.103,dotphi)
15 substitute (dotphi, DgijDt)        # cdb (ex-15-02.104,dotphi)
16 substitute (dotphi, gdotK)         # cdb (ex-15-02.105,dotphi)
17 map_sympy (dotphi, "simplify")     # cdb (ex-15-02.106,dotphi)
18
19 DphiDt    := \partial_t\{\phi\} -> @(dotphi).
20
21 checkpoint.append (dotphi)
```

$$\frac{d\phi}{dt} = \frac{1}{12} \partial_t (\log(g)) \quad (\text{ex-15-02.101})$$

$$= \frac{1}{12} g^{-1} \partial_t g \quad (\text{ex-15-02.102})$$

$$= \frac{1}{12} g^{-1} g g^{ij} \partial_t g_{ij} \quad (\text{ex-15-02.103})$$

$$= -\frac{1}{6} g^{-1} g g^{ij} N K_{ij} \quad (\text{ex-15-02.104})$$

$$= -\frac{1}{6} g^{-1} g K N \quad (\text{ex-15-02.105})$$

$$= -\frac{1}{6} K N \quad (\text{ex-15-02.106})$$

## 15.2 Evolution equation for $\bar{g}_{ij}$

```
1  gBarij := gBar_{i j} -> \exp(-4\phi) g_{i j}.
2  Kij     := K_{i j} -> A_{i j} + (1/3) g_{i j} trK.
3  A2ABar := \exp(-4\phi) A_{i j} -> ABar_{i j}.
4  ABar2A := ABar_{i j} -> \exp(-4\phi) A_{i j}.
5
6  dotgBarij := \partial_t\{gBar_{i j}\}.
7
8  substitute (dotgBarij, gBarij)          # cdb (ex-15-03.101,dotgBarij)
9  product_rule (dotgBarij)                # cdb (ex-15-03.102,dotgBarij)
10 substitute (dotgBarij, dexp)             # cdb (ex-15-03.103,dotgBarij)
11 substitute (dotgBarij, DgijDt)           # cdb (ex-15-03.104,dotgBarij)
12 substitute (dotgBarij, DphiDt)           # cdb (ex-15-03.105,dotgBarij)
13 substitute (dotgBarij, Kij)              # cdb (ex-15-03.106,dotgBarij)
14 distribute (dotgBarij)                  # cdb (ex-15-03.107,dotgBarij)
15 map_sympy (dotgBarij, "simplify")        # cdb (ex-15-03.108,dotgBarij)
16 substitute (dotgBarij, A2ABar)           # cdb (ex-15-03.109,dotgBarij)
17
18 DgBarijDt := \partial_t\{gBar_{i j}\} -> @(dotgBarij).
19
20 checkpoint.append (dotgBarij)
```

$$\frac{d\bar{g}_{ij}}{dt} = \partial_t (\exp(-4\phi) g_{ij}) \quad (\text{ex-15-03.101})$$

$$= \partial_t (\exp(-4\phi)) g_{ij} + \exp(-4\phi) \partial_t g_{ij} \quad (\text{ex-15-03.102})$$

$$= -4 \exp(-4\phi) \partial_t \phi g_{ij} + \exp(-4\phi) \partial_t g_{ij} \quad (\text{ex-15-03.103})$$

$$= -4 \exp(-4\phi) \partial_t \phi g_{ij} - 2 \exp(-4\phi) N K_{ij} \quad (\text{ex-15-03.104})$$

$$= \frac{2}{3} \exp(-4\phi) K N g_{ij} - 2 \exp(-4\phi) N K_{ij} \quad (\text{ex-15-03.105})$$

$$= \frac{2}{3} \exp(-4\phi) K N g_{ij} - 2 \exp(-4\phi) N \left( A_{ij} + \frac{1}{3} g_{ij} K \right) \quad (\text{ex-15-03.106})$$

$$= \frac{2}{3} \exp(-4\phi) K N g_{ij} - 2 \exp(-4\phi) N A_{ij} - \frac{2}{3} \exp(-4\phi) N g_{ij} K \quad (\text{ex-15-03.107})$$

$$= -2N \exp(-4\phi) A_{ij} \quad (\text{ex-15-03.108})$$

$$= -2N \bar{A}_{ij} \quad (\text{ex-15-03.109})$$