Example 3a The Riemann curvature tensor

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\partial{#}::PartialDerivative.

\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2});

;::Symbol;  # Suggsted by Kasper as a way to make use of ; legal
            # see https://cadabra.science/qa/473/is-this-legal-syntax
            # this code works with and without this trick

# rules for the first two covariant derivs of V^a

deriv1 := V^{a}_{; b}      -> \partial_{b}{V^{a}}
                            + \Gamma^{a}_{c b} V^{c}.        # cdb (ex-03.101,deriv1)

deriv2 := V^{a}_{; b ; c}  -> \partial_{c}{V^{a}_{; b}}
                            + \Gamma^{a}_{d c} V^{d}_{; b}
                            - \Gamma^{d}_{b c} V^{a}_{; d}.  # cdb (ex-03.102,deriv2)

substitute (deriv2,deriv1)                  # cdb (ex-03.103, deriv2)

Vabc := V^{a}_{; b ; c} - V^{a}_{; c ; b}.   # cdb (ex-03.104, Vabc)

substitute (Vabc,deriv2)                    # cdb (ex-03.105, Vabc)

distribute     (Vabc)                       # cdb (ex-03.106, Vabc)
product_rule   (Vabc)                       # cdb (ex-03.107, Vabc)

sort_product   (Vabc)                       # cdb (ex-03.108, Vabc)
rename_dummies (Vabc)                       # cdb (ex-03.109, Vabc)
canonicalise   (Vabc)                       # cdb (ex-03.110, Vabc)

sort_sum       (Vabc)                       # cdb (ex-03.111, Vabc)
factor_out     (Vabc,$V^{a?}$)              # cdb (ex-03.112, Vabc)

checkpoint.append (Vabc)
```

```

# create rule for Riemann, export later (for use by lib/dgeom)

substitute (Vabc,$V^{a} -> -1$)                # cdb (ex-03.113, Vabc)
                                               # note use of -1 to get correct
                                               # signs when coupled with the rule
                                               # for Rabcd (next statement)

Rabcd := R^{a}_{d b c} -> @(Vabc).             # cdb (ex-03.114, Rabcd) #

foo   := R^{a}_{b c d}.                        # cdb (ex-03.115, foo)
substitute (foo, Rabcd)                        # cdb (ex-03.116, foo)

# update rule to use nice indices

Rabcd := R^{a}_{b c d} -> @(foo).

checkpoint.append (Rabcd)
```

$$V^{a}{}_{;b} \rightarrow \partial_{b}V^{a} + \Gamma^{a}{}_{cb}V^{c} \tag{ex-03.101}$$

$$V^{a}{}_{;b;c} \rightarrow \partial_{c}V^{a}{}_{;b} + \Gamma^{a}{}_{dc}V^{d}{}_{;b} - \Gamma^{d}{}_{bc}V^{a}{}_{;d} \tag{ex-03.102}$$

$$V^{a}{}_{;b;c} \rightarrow \partial_{c}\left(\partial_{b}V^{a} + \Gamma^{a}{}_{db}V^{d}\right) + \Gamma^{a}{}_{dc}\left(\partial_{b}V^{d} + \Gamma^{d}{}_{eb}V^{e}\right) - \Gamma^{d}{}_{bc}\left(\partial_{d}V^{a} + \Gamma^{a}{}_{ed}V^{e}\right) \tag{ex-03.103}$$

$$
\begin{aligned}
V^a{}_{;b;c} - V^a{}_{;c;b} &= \partial_c\big(\partial_b V^a + \Gamma^a{}_{db}V^d\big) + \Gamma^a{}_{dc}\left(\partial_b V^d + \Gamma^d{}_{eb}V^e\right) - \Gamma^d{}_{bc}\left(\partial_d V^a + \Gamma^a{}_{ed}V^e\right) - \partial_b\big(\partial_c V^a + \Gamma^a{}_{dc}V^d\big) - \Gamma^a{}_{db}\left(\partial_c V^d + \Gamma^d{}_{ec}V^e\right) \\
&\quad + \Gamma^d{}_{cb}\left(\partial_d V^a + \Gamma^a{}_{ed}V^e\right) && \text{(ex-03.105)} \\
&= \partial_{cb}V^a + \partial_c\big(\Gamma^a{}_{db}V^d\big) + \Gamma^a{}_{dc}\partial_b V^d + \Gamma^a{}_{dc}\Gamma^d{}_{eb}V^e - \Gamma^d{}_{bc}\partial_d V^a - \Gamma^d{}_{bc}\Gamma^a{}_{ed}V^e - \partial_{bc}V^a - \partial_b\big(\Gamma^a{}_{dc}V^d\big) - \Gamma^a{}_{db}\partial_c V^d - \Gamma^a{}_{db}\Gamma^d{}_{ec}V^e + \Gamma^d{}_{cb}\partial_d V^a \\
&\quad + \Gamma^d{}_{cb}\Gamma^a{}_{ed}V^e && \text{(ex-03.106)} \\
&= \partial_{cb}V^a + \partial_c\Gamma^a{}_{db}V^d + \Gamma^a{}_{dc}\Gamma^d{}_{eb}V^e - \Gamma^d{}_{bc}\partial_d V^a - \Gamma^d{}_{bc}\Gamma^a{}_{ed}V^e - \partial_{bc}V^a - \partial_b\Gamma^a{}_{dc}V^d - \Gamma^a{}_{db}\Gamma^d{}_{ec}V^e + \Gamma^d{}_{cb}\partial_d V^a + \Gamma^d{}_{cb}\Gamma^a{}_{ed}V^e && \text{(ex-03.107)} \\
&= \partial_{cb}V^a + V^d\partial_c\Gamma^a{}_{db} + V^e\Gamma^a{}_{dc}\Gamma^d{}_{eb} - \Gamma^d{}_{bc}\partial_d V^a - V^e\Gamma^a{}_{ed}\Gamma^d{}_{bc} - \partial_{bc}V^a - V^d\partial_b\Gamma^a{}_{dc} - V^e\Gamma^a{}_{db}\Gamma^d{}_{ec} + \Gamma^d{}_{cb}\partial_d V^a + V^e\Gamma^a{}_{ed}\Gamma^d{}_{cb} && \text{(ex-03.108)} \\
&= \partial_{cb}V^a + V^d\partial_c\Gamma^a{}_{db} + V^d\Gamma^a{}_{ec}\Gamma^e{}_{db} - \Gamma^d{}_{bc}\partial_d V^a - V^d\Gamma^a{}_{de}\Gamma^e{}_{bc} - \partial_{bc}V^a - V^d\partial_b\Gamma^a{}_{dc} - V^d\Gamma^a{}_{eb}\Gamma^e{}_{dc} + \Gamma^d{}_{cb}\partial_d V^a + V^d\Gamma^a{}_{de}\Gamma^e{}_{cb} && \text{(ex-03.109)} \\
&= V^d\partial_c\Gamma^a{}_{bd} + V^d\Gamma^a{}_{ce}\Gamma^e{}_{bd} - V^d\partial_b\Gamma^a{}_{cd} - V^d\Gamma^a{}_{be}\Gamma^e{}_{cd} && \text{(ex-03.110)} \\
&= V^d\partial_c\Gamma^a{}_{bd} - V^d\partial_b\Gamma^a{}_{cd} - V^d\Gamma^a{}_{be}\Gamma^e{}_{cd} + V^d\Gamma^a{}_{ce}\Gamma^e{}_{bd} && \text{(ex-03.111)} \\
&= V^d\left(\partial_c\Gamma^a{}_{bd} - \partial_b\Gamma^a{}_{cd} - \Gamma^a{}_{be}\Gamma^e{}_{cd} + \Gamma^a{}_{ce}\Gamma^e{}_{bd}\right) && \text{(ex-03.112)} \\
&= -R^a{}_{dbc}V^d \quad \text{(MTW)}
\end{aligned}
$$

$$
R^a{}_{bcd} = -\,\partial_d\Gamma^a{}_{cb} + \partial_c\Gamma^a{}_{db} + \Gamma^a{}_{ce}\Gamma^e{}_{db} - \Gamma^a{}_{de}\Gamma^e{}_{cb}
$$

Example 3b The Riemann curvature tensor

This differs from the above by not using the ::TableauSymmetry property. It gives the same results as above but it does require a little bit more housekeeping.

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\partial{#}::PartialDerivative.

;::Symbol;  # Suggsted by Kasper as a way to make use of ; legal
            # see https://cadabra.science/qa/473/is-this-legal-syntax
            # this code works with and without this trick

# rules for the first two covariant derivs of V^a

deriv1 := V^{a}_{; b}      -> \partial_{b}{V^{a}}
                            + \Gamma^{a}_{c b} V^{c}.        # cdb (ex-03.301,deriv1)

deriv2 := V^{a}_{; b ; c}  -> \partial_{c}{V^{a}_{; b}}
                            + \Gamma^{a}_{d c} V^{d}_{; b}
                            - \Gamma^{d}_{b c} V^{a}_{; d}.  # cdb (ex-03.302,deriv2)

substitute (deriv2,deriv1)                     # cdb (ex-03.303, deriv2)

Vabc := V^{a}_{; b ; c} - V^{a}_{; c ; b}.    # cdb (ex-03.304, Vabc)

substitute (Vabc,deriv2)                       # cdb (ex-03.305, Vabc)

distribute     (Vabc)                          # cdb (ex-03.306, Vabc)
product_rule   (Vabc)                          # cdb (ex-03.307, Vabc)

# -----------------------------------------------------
# trick to obtain a symmetric connection

G_{a b}::Symmetric.

substitute     (Vabc,$\Gamma^{a}_{b c} ->  G^{a} G_{b c}$)
sort_product   (Vabc)                          # cdb (ex-03.308, Vabc)
```

```
rename_dummies (Vabc)                              # cdb (ex-03.309, Vabc)
canonicalise   (Vabc)                              # cdb (ex-03.310, Vabc)
substitute     (Vabc,$G^{a} G_{b c} -> \Gamma^{a}_{b c}$,repeat=True)
# ------------------------------------------------------

sort_product   (Vabc)
rename_dummies (Vabc)
canonicalise   (Vabc)

sort_sum       (Vabc)                              # cdb (ex-03.311, Vabc)
factor_out     (Vabc,$V^{a?}$)                     # cdb (ex-03.312, Vabc)

checkpoint.append (Vabc)
```

$$V^{a}{}_{;b} \rightarrow \partial_{b}V^{a} + \Gamma^{a}{}_{cb}V^{c} \tag{ex-03.301}$$

$$V^{a}{}_{;b;c} \rightarrow \partial_{c}V^{a}{}_{;b} + \Gamma^{a}{}_{dc}V^{d}{}_{;b} - \Gamma^{d}{}_{bc}V^{a}{}_{;d} \tag{ex-03.302}$$

$$V^{a}{}_{;b;c} \rightarrow \partial_{c}\left(\partial_{b}V^{a} + \Gamma^{a}{}_{db}V^{d}\right) + \Gamma^{a}{}_{dc}\left(\partial_{b}V^{d} + \Gamma^{d}{}_{eb}V^{e}\right) - \Gamma^{d}{}_{bc}\left(\partial_{d}V^{a} + \Gamma^{a}{}_{ed}V^{e}\right) \tag{ex-03.303}$$

$$
\begin{aligned}
V^a{}_{;b;c} - V^a{}_{;c;b} &= \partial_c\big(\partial_b V^a + \Gamma^a{}_{db}V^d\big) + \Gamma^a{}_{dc}\left(\partial_b V^d + \Gamma^d{}_{eb}V^e\right) - \Gamma^d{}_{bc}\left(\partial_d V^a + \Gamma^a{}_{ed}V^e\right) - \partial_b\big(\partial_c V^a + \Gamma^a{}_{dc}V^d\big) - \Gamma^a{}_{db}\left(\partial_c V^d + \Gamma^d{}_{ec}V^e\right) \\
&\quad + \Gamma^d{}_{cb}\left(\partial_d V^a + \Gamma^a{}_{ed}V^e\right) && \text{(ex-03.305)} \\
&= \partial_{cb}V^a + \partial_c\big(\Gamma^a{}_{db}V^d\big) + \Gamma^a{}_{dc}\partial_b V^d + \Gamma^a{}_{dc}\Gamma^d{}_{eb}V^e - \Gamma^d{}_{bc}\partial_d V^a - \Gamma^d{}_{bc}\Gamma^a{}_{ed}V^e - \partial_{bc}V^a - \partial_b\big(\Gamma^a{}_{dc}V^d\big) - \Gamma^a{}_{db}\partial_c V^d - \Gamma^a{}_{db}\Gamma^d{}_{ec}V^e + \Gamma^d{}_{cb}\partial_d V^a \\
&\quad + \Gamma^d{}_{cb}\Gamma^a{}_{ed}V^e && \text{(ex-03.306)} \\
&= \partial_{cb}V^a + \partial_c\Gamma^a{}_{db}V^d + \Gamma^a{}_{dc}\Gamma^d{}_{eb}V^e - \Gamma^d{}_{bc}\partial_d V^a - \Gamma^d{}_{bc}\Gamma^a{}_{ed}V^e - \partial_{bc}V^a - \partial_b\Gamma^a{}_{dc}V^d - \Gamma^a{}_{db}\Gamma^d{}_{ec}V^e + \Gamma^d{}_{cb}\partial_d V^a + \Gamma^d{}_{cb}\Gamma^a{}_{ed}V^e && \text{(ex-03.307)} \\
&= \partial_{cb}V^a + V^d\partial_c(G^aG_{db}) + G^aG^dG_{dc}G_{eb}V^e - G^dG_{bc}\partial_d V^a - G^aG^dG_{bc}G_{ed}V^e - \partial_{bc}V^a - V^d\partial_b(G^aG_{dc}) - G^aG^dG_{db}G_{ec}V^e \\
&\quad + G^dG_{cb}\partial_d V^a + G^aG^dG_{cb}G_{ed}V^e && \text{(ex-03.308)} \\
&= \partial_{cb}V^a + V^d\partial_c(G^aG_{db}) + G^aG^dG_{dc}G_{eb}V^e - G^dG_{bc}\partial_d V^a - G^aG^dG_{bc}G_{ed}V^e - \partial_{bc}V^a - V^d\partial_b(G^aG_{dc}) - G^aG^dG_{db}G_{ec}V^e \\
&\quad + G^dG_{cb}\partial_d V^a + G^aG^dG_{cb}G_{ed}V^e && \text{(ex-03.309)} \\
&= V^d\partial_c(G^aG_{bd}) + G^aG^dG_{be}G_{cd}V^e - V^d\partial_b(G^aG_{cd}) - G^aG^dG_{bd}G_{ce}V^e && \text{(ex-03.310)} \\
&= V^d\partial_c\Gamma^a{}_{bd} - V^d\partial_b\Gamma^a{}_{cd} + V^d\Gamma^a{}_{bd}\Gamma^e{}_{ce} - V^d\Gamma^a{}_{be}\Gamma^e{}_{cd} && \text{(ex-03.311)} \\
&= V^d\left(\partial_c\Gamma^a{}_{bd} - \partial_b\Gamma^a{}_{cd} + \Gamma^a{}_{bd}\Gamma^e{}_{ce} - \Gamma^a{}_{be}\Gamma^e{}_{cd}\right) && \text{(ex-03.312)} \\
&= -R^a{}_{dbc}V^d \quad \text{(MTW)}
\end{aligned}
$$