Example 6-01 Evaluating components

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

\partial{#}::PartialDerivative.

V  := { V_{\theta} = \varphi, V_{\varphi} = \sin(\theta) }.   # cdb(ex-06.100,V)
dV := \partial_{b}{V_{a}} - \partial_{a}{V_{b}}.             # cdb(ex-06.101,dV)

evaluate (dV, V)   # cdb(ex-06.102,dV)
```

$$V_a = [V_\theta = \varphi,\ V_\varphi = \sin\theta] \tag{ex-06.100}$$

$$\partial_b V_a - \partial_a V_b = \Box_{ab} \begin{cases} \Box_{\varphi\theta} = \cos\theta - 1 \\ \Box_{\theta\varphi} = -\cos\theta + 1 \end{cases} \tag{ex-06.102}$$

Example 6-02 Ricci tensor of a 2-sphere

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

\partial{#}::PartialDerivative.

Gamma := \Gamma^{a}_{f g} -> 1/2 g^{a b} (   \partial_{g}{g_{b f}}
                                           + \partial_{f}{g_{b g}}
                                           - \partial_{b}{g_{f g}} ).

Rabcd := R^{d}_{e f g} ->   \partial_{f}{\Gamma^{d}_{e g}}
                          - \partial_{g}{\Gamma^{d}_{e f}}
                          + \Gamma^{d}_{b f} \Gamma^{b}_{e g}
                          - \Gamma^{d}_{b g} \Gamma^{b}_{e f}.

Rab := R_{a b} -> R^{c}_{a c b}.

gab := { g_{\theta\theta}   = r**2,
         g_{\varphi\varphi} = r**2 \sin(\theta)**2 }.      # cdb(ex-06.201,gab)

iab := { g^{\theta\theta}   = 1/r**2,
         g^{\varphi\varphi} = 1/(r**2 \sin(\theta)**2) }.  # cdb(ex-06.202,iab)

substitute (Rabcd, Gamma)
substitute (Rab, Rabcd)

evaluate   (Gamma, gab+iab, rhsonly=True)                  # cdb(ex-06.203,Gamma)
evaluate   (Rabcd, gab+iab, rhsonly=True)                  # cdb(ex-06.204,Rabcd)
evaluate   (Rab,   gab+iab, rhsonly=True)                  # cdb(ex-06.205,Rab)

Riem := R^{d}_{e f g}.
substitute (Riem, Rabcd)
evaluate   (Riem, gab+iab)                                 # cdb(ex-06.206,Riem)

expr := R_{a b}.
substitute (expr, Rab)
evaluate   (expr, gab+iab)
```

```
# cdbBeg(print.ex-06.02)
print ('Rab = ' + str(Rab.input_form())+'\n')  # reveals Cadabra's internal structure for storing Rab

print ('Rab[0] = ' + str(Rab[0]))
print ('Rab[1] = ' + str(Rab[1])+'\n')

print ('Rab[1][0] = ' + str(Rab[1][0]))
print ('Rab[1][1] = ' + str(Rab[1][1]))
print ('Rab[1][2] = ' + str(Rab[1][2])+'\n')

print ('Rab[1][2][0] = ' + str(Rab[1][2][0]))
print ('Rab[1][2][0][0] = ' + str(Rab[1][2][0][0]))
print ('Rab[1][2][0][1] = ' + str(Rab[1][2][0][1]))
# cdbEnd(print.ex-06.02)
```

$$\left[g_{\theta\theta} = r^{2},\ g_{\varphi\varphi} = r^{2}(\sin\theta)^{2}\right] \tag{ex-06.201}$$

$$\left[g^{\theta\theta} = r^{-2},\ g^{\varphi\varphi} = \left(r^{2}(\sin\theta)^{2}\right)^{-1}\right] \tag{ex-06.202}$$

$$\Gamma^{a}{}_{fg} \rightarrow \square_{fg}{}^{a} \begin{cases} \square_{\varphi\theta}{}^{\varphi} = (\tan\theta)^{-1} \\ \square_{\theta\varphi}{}^{\varphi} = (\tan\theta)^{-1} \\ \square_{\varphi\varphi}{}^{\theta} = -\dfrac{1}{2}\sin(2\theta) \end{cases} \tag{ex-06.203}$$

$$R^{d}{}_{efg} \rightarrow \square_{eg}{}^{d}{}_{f} \begin{cases} \square_{\varphi\varphi}{}^{\theta}{}_{\theta} = (\sin\theta)^{2} \\ \square_{\theta\varphi}{}^{\varphi}{}_{\theta} = -1 \\ \square_{\varphi\theta}{}^{\theta}{}_{\varphi} = -(\sin\theta)^{2} \\ \square_{\theta\theta}{}^{\varphi}{}_{\varphi} = 1 \end{cases} \tag{ex-06.204}$$

$$R_{ab} \rightarrow \square_{ab} \begin{cases} \square_{\varphi\varphi} = (\sin\theta)^{2} \\ \square_{\theta\theta} = 1 \end{cases} \tag{ex-06.205}$$

$$\square_{eg}{}^{d}{}_{f} \begin{cases} \square_{\varphi\varphi}{}^{\theta}{}_{\theta} = (\sin\theta)^{2} \\ \square_{\theta\varphi}{}^{\varphi}{}_{\theta} = -1 \\ \square_{\varphi\theta}{}^{\theta}{}_{\varphi} = -(\sin\theta)^{2} \\ \square_{\theta\theta}{}^{\varphi}{}_{\varphi} = 1 \end{cases} \tag{ex-06.206}$$

```
Rab = R_{a b} -> \components_{a b}({{\varphi, \varphi} = (\sin(\theta))**2, {\theta, \theta} = 1})

Rab[0] = R_{a b}
Rab[1] = \components_{a b}({{\varphi, \varphi} = (\sin(\theta))**2, {\theta, \theta} = 1})

Rab[1][0] = a
Rab[1][1] = b
Rab[1][2] = {{\varphi, \varphi} = (\sin(\theta))**2, {\theta, \theta} = 1}

Rab[1][2][0] = {\varphi, \varphi} = (\sin(\theta))**2
Rab[1][2][0][0] = {\varphi, \varphi}
Rab[1][2][0][1] = (\sin(\theta))**2
```

Example 6-03 Using *complete* to compute the inverse metric

This version uses `complete` to compute the inverse metric.

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

\partial{#}::PartialDerivative.

g^{a b}::InverseMetric.  # essential when using complete (gab, $g^{a b}$)

Gamma := \Gamma^{a}_{f g} -> 1/2 g^{a b} (   \partial_{g}{g_{b f}}
                                          + \partial_{f}{g_{b g}}
                                          - \partial_{b}{g_{f g}} ).

Rabcd := R^{d}_{e f g} ->   \partial_{f}{\Gamma^{d}_{e g}}
                          - \partial_{g}{\Gamma^{d}_{e f}}
                          + \Gamma^{d}_{b f} \Gamma^{b}_{e g}
                          - \Gamma^{d}_{b g} \Gamma^{b}_{e f}.

Rab := R_{a b} -> R^{c}_{a c b}.

gab := { g_{\theta\theta}   = r**2,
         g_{\varphi\varphi} = r**2 \sin(\theta)**2 }.      # cdb(ex-06.301,gab)

complete   (gab, $g^{a b}$)                                # cdb(ex-06.302,gab)

substitute (Rabcd, Gamma)
substitute (Rab, Rabcd)

evaluate   (Gamma, gab, rhsonly=True)                      # cdb(ex-06.303,Gamma)
evaluate   (Rabcd, gab, rhsonly=True)                      # cdb(ex-06.304,Rabcd)
evaluate   (Rab,   gab, rhsonly=True)                      # cdb(ex-06.305,Rab)

Riem := R^{d}_{e f g}.
substitute (Riem, Rabcd)
evaluate   (Riem, gab)                                     # cdb(ex-06.306,Riem)
```

$$\left[g_{\theta\theta} = r^2,\ g_{\varphi\varphi} = r^2(\sin\theta)^2\right] \tag{ex-06.301}$$

$$\left[g_{\theta\theta} = r^2,\ g_{\varphi\varphi} = r^2(\sin\theta)^2,\ g^{\theta\theta} = r^{-2},\ g^{\varphi\varphi} = \left(r^2(\sin\theta)^2\right)^{-1}\right] \tag{ex-06.302}$$

$$\Gamma^{a}{}_{fg} \rightarrow \Box_{fg}{}^{a} \begin{cases} \Box_{\varphi\theta}{}^{\varphi} = (\tan\theta)^{-1} \\ \Box_{\theta\varphi}{}^{\varphi} = (\tan\theta)^{-1} \\ \Box_{\varphi\varphi}{}^{\theta} = -\dfrac{1}{2}\sin(2\theta) \end{cases} \tag{ex-06.303}$$

$$R^{d}{}_{efg} \rightarrow \Box_{eg}{}^{d}{}_{f} \begin{cases} \Box_{\varphi\varphi}{}^{\theta}{}_{\theta} = (\sin\theta)^2 \\ \Box_{\theta\varphi}{}^{\varphi}{}_{\theta} = -1 \\ \Box_{\varphi\theta}{}^{\theta}{}_{\varphi} = -(\sin\theta)^2 \\ \Box_{\theta\theta}{}^{\varphi}{}_{\varphi} = 1 \end{cases} \tag{ex-06.304}$$

$$R_{ab} \rightarrow \Box_{ab} \begin{cases} \Box_{\varphi\varphi} = (\sin\theta)^2 \\ \Box_{\theta\theta} = 1 \end{cases} \tag{ex-06.305}$$

$$\Box_{eg}{}^{d}{}_{f} \begin{cases} \Box_{\varphi\varphi}{}^{\theta}{}_{\theta} = (\sin\theta)^2 \\ \Box_{\theta\varphi}{}^{\varphi}{}_{\theta} = -1 \\ \Box_{\varphi\theta}{}^{\theta}{}_{\varphi} = -(\sin\theta)^2 \\ \Box_{\theta\theta}{}^{\varphi}{}_{\varphi} = 1 \end{cases} \tag{ex-06.306}$$

Example 6-04 Components by scalar projection

This example shows how one component of the metric tensor can be computed using a scalar projection.

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

theta{#}::LaTeXForm{"\theta"}.
varphi{#}::LaTeXForm{"\varphi"}.

gab := { g_{\theta \theta}   = r**2,
         g_{\varphi \varphi} = r**2 \sin(\theta)**2 }.   # cdb(ex-06.400,gab)

# obtain components by contracting with basis vectors

basis := {theta^{\theta} = 1, varphi^{\varphi} = 1}.

compt := g_{a b} varphi^{a} varphi^{b}.                  # cdb(ex-06.401,compt)

evaluate (compt,gab+basis)                               # cdb(ex-06.402,compt)

gphiphi = compt._sympy_()

# cdbBeg(print.ex-06.04)
print ('type compt   = ' + str(type(compt)))      # shows that compt is a Cadabra object
print ('type gphiphi = ' + str(type(gphiphi)))    # shows that gphiphi is a Python object
print ('     compt   = ' + str(compt))            # will contain LaTeX markup
print ('     gphiphi = ' + str(gphiphi))          # will be pure Python/SymPy
# cdbEnd(print.ex-06.04)

json.append (compt)
```

$$
\begin{aligned}
g_{\varphi\varphi} &= g_{ab}\varphi^{a}\varphi^{b} && \texttt{(ex-06.401)}\\
&= r^{2}(\sin\theta)^{2} && \texttt{(ex-06.402)}
\end{aligned}
$$

```
type compt   = <class 'cadabra2.Ex'>
type gphiphi = <class 'sympy.core.mul.Mul'>
     compt   = (r)**2 (\sin(\theta))**2
     gphiphi = r**2*sin(theta)**2
```

Example 6-05 Components by selection

This example shows how one component of the metric tensor can be computed by indexing the result of a call to `evaluate`.

```
{\theta, \varphi}::Coordinate.
{a,b,c,d,e,f,g,h#}::Indices(values={\theta, \varphi}, position=independent).

theta{#}::LaTeXForm{"\theta"}.
varphi{#}::LaTeXForm{"\varphi"}.

gab := { g_{\theta \theta}   = r**2,
         g_{\varphi \varphi} = r**2 \sin(\theta)**2 }.   # cdb(ex-06.500,gab)

metric := g_{a b}.

evaluate (metric,gab)

indcs = metric[2][1][0]                                  # cdb(ex-06.501,indcs)
compt = metric[2][1][1]                                  # cdb(ex-06.502,compt)

# cdbBeg(print.ex-06.05)
print ('metric = ' + str(metric.input_form())+'\n')  # reveals Cadabra's internal structure for storing metric

print ('metric[0] = ' + str(metric[0]))
print ('metric[1] = ' + str(metric[1]))
print ('metric[2] = ' + str(metric[2])+'\n')

print ('metric[2][1] = '+ str(metric[2][1]))
print ('metric[2][1][0] = '+ str(metric[2][1][0]))
print ('metric[2][1][1] = '+ str(metric[2][1][1]))
# cdbEnd(print.ex-06.05)

json.append (indcs)
json.append (compt)
```

$$g_{\varphi\varphi} = g_{[\varphi,\ \varphi]} \tag{ex-06.501}$$

$$= r^2(\sin\theta)^2 \tag{ex-06.502}$$

```
metric = \components_{a b}({{\theta, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2})

metric[0] = a
metric[1] = b
metric[2] = {{\theta, \theta} = (r)**2, {\varphi, \varphi} = (r)**2 (\sin(\theta))**2}

metric[2][1] = {\varphi, \varphi} = (r)**2 (\sin(\theta))**2
metric[2][1][0] = {\varphi, \varphi}
metric[2][1][1] = (r)**2 (\sin(\theta))**2
```